JONATHAN CHOATE

TETRAHEDRAL TREATS

Have you ever wondered about the geometry of tetrahedra? It would seem that there should be some fascinating results, like those for a triangle, but you never see any reference to them. Ross Honsberger offers an explanation for this lack of information in his book *Mathematical Gems II* [Mathematical Assn of America, 1976, ISBN 0883853027]:

> *"Solid geometry pays as much attention to the tetrahedron as plane geometry does to the triangle. Yet many elementary properties of the tetrahedron are not very well known.*
>
> *Solid geometry is often a more complicated subject than plane geometry, for it is undoubtedly more difficult for the mind's eye to establish and maintain a constant picture of the relevant positions of objects in three-dimensional space. Flat figures are much easier to think about and to describe to others. Thus the pursuit of solid geometry demands greater motivation."*

Computer algebra systems and 3D construction software make the study of solid geometry accessible to more students. A TI-89 graphing calculator [or Mathematica, which I use in this article since the visual display of Mathematica is cleaner than that of the TI-89] allows quick computations that would be tedious and time-consuming if done by hand. This Geometer's Corner is about how you can use Cabri 3D, a 3D construction program, and a computer algebra system such as that on a TI-89 to study tetrahedra. As I mentioned earlier it should come as no surprise that many of the special properties of triangles have 3D equivalents. In what follows we will look for the 3D equivalents of the four special points that every triangle has.

- A centroid (G): the intersection of the medians. G is located two thirds of the way from a vertex to the midpoint of the opposite side;
- A circumcenter (O): the intersection of the perpendicular bisectors of the three sides and the center of the circumscribed circle;
- An incenter (I): the intersection of the angle bisectors and the center of the inscribed circle;
- An orthocenter (H): the intersection of the altitudes.

In addition O, G, and H all lie on the Euler line in such a way that $GH = 2OG$.



FIGURE 1. TRIANGLE *ABC* WITH ORTHOCENTER *H*, CENTROID *G*, INCENTER *I*, CIRCUMCENTER *O*, AND EULER LINE *OH*.

Tetrahedra have equivalent centers, although not every tetrahedron has an orthocenter. We will use Mathematica to find each of these special points for the tetrahedron *ABCD* with vertices at $A = \{2, 4, 0\}$, $B = \{6, 8, 0\}$, $C = \{8, -2, 0\}$, and $D = \{4, 2, 10\}$. If you are interested in proofs of the results, you can find them in the Tetrahedral Geometry section of my Website, www.zebragraph.com.

1. Finding the Centroid

A tetrahedron's median is a segment connecting a vertex and the centroid of the opposite face. The image of tetrahedron *ABCD* with three of its four medians in **Figure 2** was constructed using Cabri 3D. The medians certainly appear to intersect and in an interesting way.

Send solutions to old problems and any new ideas to the Geometer's Corner editor: Jonathan Choate, Groton School, Box 991, Groton, MA 01450.



FIGURE 2. TETRAHEDRON *ABCD* WITH CENTROID *G*. G_1 IS THE CENTROID OF FACE *BDC*; G_2 IS THE CENTROID OF FACE *ACD*.

The median AG_1 from vertex *A* to face *BCD* can be found by first finding the centroid G_1 of face *BCD*.

$OG_1 = 1/3(OB + OC + OD)$

Line AG_1 has vector equation

$(1 - t)OA + tOG_1$.

The median BG_2 from *B* to face *ACD* where

$OG_2 = 1/3(OA + OC + OD)$

has vector equation

$(1 - s)OB + sOG_2$.

To find the point of intersection you need to solve the equation

$(1 - t)OA + tOG_1 = (1 - s)OB + sOG_2$

for *s* and *t*. Here is a solution using Mathematica. In what follows *OP* denotes the vector from the origin *O* to point *P*. This creates a slight problem later because the circumcenter of a triangle has traditionally been labeled *O*, and so we use the label *R* for the center of the circumsphere. (See **Table 1**.)

That *s* and *t* both equal 3/4 shows that the medians intersect in a special way. $AG = 3/4AG_1$ and $BG = 3/4BG_2$. Put another way, the centroid divides each median in the ratio 1:3. This was first proved in 1565 by the Italian mathematician Fredrico Commandino and appeared as Prop. 17 in his *De Centro Gravitates Solidorum*.

Finding the Centroid

```
OA = {2, 4, 0}; OB = {6, 8, 0}; OC = {8, -2, 0};
OD = {4, 2, 10};

OG1 = (1/3) (OB + OC + OD)

{6, 8/3, 10/3}

OG2 = (1/3) (OA + OC + OD)

{14/3, 4/3, 10/3}

medianAG1 = (1 - t) OA + t OG1

{2 (1 - t) + 6 t, 4 (1 - t) + 8 t/3, 10 t/3}

medianAG2 = (1 - s) OB + s OG2

{6 (1 - s) + 14 s/3, 8 (1 - s) + 4 s/3, 10 s/3}

Solve[medianAG1 == medianAG2, {s, t}]

{{s -> 3/4, t -> 3/4}}

s = 3/4

3/4

centroid = medianAG2

{5, 3, 5/2}
```

TABLE 1.

Commandino's Theorem

The four medians of a tetrahedron concur in a point that divides each of them in the ratio 1:3, the longer segment being on the side of the vertex of the tetrahedron.

A proof of Commandino's Theorem can be found at www.zebragraph.com/3DGems.html

In order to find the circumcenter and the incenter you need to find the intersection of three planes. This is where a computer algebra system comes in handy. All computer algebra systems can calculate both the dot product $\boldsymbol{u} \bullet \boldsymbol{v}$ and the cross product $\boldsymbol{u} \times \boldsymbol{v}$ of two three-dimensional vectors $\boldsymbol{u}$ and $\boldsymbol{v}$. In what follows, we will make use of the followoing facts.

- If $\boldsymbol{u} \bullet \boldsymbol{v} = 0$ then $\boldsymbol{u}$ and $\boldsymbol{v}$ are perpendicular.
- Let $\boldsymbol{u}$ be a vector perpendicular to plane *P*. If $X = \{x, y, z\}$ is any point in plane *P* and $A = \{x_A, y_A, z_A\}$ is a particular point in the plane, then $\boldsymbol{u} \bullet (OX - OA) = 0$.
- The plane through points *A*, *B*, and *C* has equation $(AB \times AC) \bullet (OX - OA) = 0$.

2. Finding the Circumcenter

To find the circumcenter R you have to find the intersection of the perpendicular bisectors of the six edges. This intersection can be found by finding the intersection of three perpendicular bisecting planes of three non-planar edges. (Convince yourself that the intersection of three bisecting planes of any three edges that form a face must be a line.) **Figure 3a** shows tetrahedron $ABCD$ with the perpendicular bisecting planes of edges AB (solid shading), BD (small holes), and AC (large holes). **Figure 3b** shows the circumscribed sphere.



FIGURE 3A. BISECTING EDGE PLANES INTERSECTING AT R, THE CIRCUMCENTER.



FIGURE 3B. TETRAHEDRON $ABCD$ WITH ITS CIRCUMSPHERE.

Finding the Circumcenter

```
OX = {x, y, z}

{x, y, z}

midAC = (1/2) (OA + OC)

{5, 1, 0}

perpbisectplaneAC = AC.(OX - midAC)

6 (-5 + x) - 6 (-1 + y)

midAB = (1/2) (OA + OB)

{4, 6, 0}

AB = OB - OA

{4, 4, 0}

perpbisectplaneAB = AB.(OX - midAB)

4 (-4 + x) + 4 (-6 + y)

midBD = (1/2) (OB + OD)

{5, 5, 5}

BD = OD - OB

{-2, -6, 10}

perpbisectplaneBD = BD.(OX - midBD)

-2 (-5 + x) - 6 (-5 + y) + 10 (-5 + z)

Solve[{perpbisectplaneAC == 0, perpbisectplaneAB == 0,
  perpbisectplaneBD == 0}, {x, y, z}]
```

$$\left\{\left\{x \to 7,\ y \to 3,\ z \to \frac{21}{5}\right\}\right\}$$

```
OR = {7, 3, 21/5}
```

$$\left\{7,\ 3,\ \frac{21}{5}\right\}$$

```
circumradius = Sqrt[(OR - OA).(OR - OA)]
```

$$\frac{\sqrt{1091}}{5}$$

TABLE 2.

3. Finding the Incenter

The incenter is located at the intersection of the six planes that bisect the six dihedral angles formed by the faces that share an edge. The task is easier if you put the equations of the planes that include a face in normal form. The equation $ax + by + cz + d = 0$ is said to be in normal form if $a^2 + b^2 + c^2 = 1$. Given two planes P_1 with normal equation $a_1x + b_1y + c_1z + d_1 = 0$ and P_2 with normal equation $a_2x + b_2y + c_2z + d_2 = 0$, the equation of the plane that bisects the dihedral angle formed by them is $(a_1 + a_2)x + (b_1 + b_2)y + (c_1 + c_2)z + (d_1 + d_2) = 0$. You actually need to find the intersection of only three of these planes, but they must be chosen so that the edges that determine the dihedral angles are not co-planar. **Figure 4a** shows the intersection I of the bisecting planes of the dihedral angles formed by faces ABC and BCD (large holes), faces ADC and BDC (small holes), and faces ABC and ABD (solid shading). **Figure 4b** shows the inscribed sphere.



FIGURE 4A. BISECTING PLANES OF DIHEDRAL ANGLES INTERSECTING AT I, THE INCENTER.



FIGURE 4B. TETRAHEDRON *ABCD* WITH ITS INSPHERE.

Finding the Incenter

```
AB = OB - OA; AC = OC - OA; BC = OC - OB; BD = OD - OB; BA = -AB;
AD = OD - OA
```

{2, -2, 10}

```
normalABC[u_, v_] := Cross[u, v] / Sqrt[Cross[u, v].Cross[u, v]]
```

```
faceABC = normalABC[AB, AC]. (OX - OA)
```

$-z$

```
faceABD = normalABC[AD, AB]. (OX - OA)
```

$$-\frac{5(-2+x)}{3\sqrt{6}} + \frac{5(-4+y)}{3\sqrt{6}} + \frac{1}{3}\sqrt{\frac{2}{3}}\,z$$

```
faceADC = normalABC[AC, AD]. (OX - OA)
```

$$-\frac{-2+x}{\sqrt{2}} - \frac{-4+y}{\sqrt{2}}$$

```
faceBDC = normalABC[BD, BC]. (OX - OB)
```

$$\frac{25(-6+x)}{\sqrt{714}} + \frac{5(-8+y)}{\sqrt{714}} + 4\sqrt{\frac{2}{357}}\,z$$

```
bisectplaneBC = faceABC + faceBDC
```

$$\frac{25(-6+x)}{\sqrt{714}} + \frac{5(-8+y)}{\sqrt{714}} - z + 4\sqrt{\frac{2}{357}}\,z$$

```
bisectplaneAC = faceABC + faceADC
```

$$-\frac{-2+x}{\sqrt{2}} - \frac{-4+y}{\sqrt{2}} - z$$

```
bisectplaneBD = faceBDC + faceABD
```

$$\frac{25(-6+x)}{\sqrt{714}} - \frac{5(-2+x)}{3\sqrt{6}} + \frac{5(-8+y)}{\sqrt{714}} + \frac{5(-4+y)}{3\sqrt{6}} + 4\sqrt{\frac{2}{357}}\,z + \frac{1}{3}\sqrt{\frac{2}{3}}\,z$$

```
Solve[{bisectplaneBC == 0, bisectplaneAC == 0, bisectplaneBD == 0},
  {x, y, z}] // N
```

{{x → 0.713153, y → 13.2779, z → -5.6505}}

```
In[93]:= incenter = {0.713153, 13.2779, -5.6505}
Out[93]= {0.713153, 13.2779, -5.6505}
```

TABLE 3.

4. Finding the Orthocenter

An altitude of a tetrahedron is a line through a vertex perpendicular to the opposite face or the plane containing the opposite face. The orthocenter is the intersection of the altitudes. As **Figure 5** shows, most tetrahedra don't have an orthocenter.



FIGURE 5. TETRAHEDRON *ABCD*, WHICH HAS NO ORTHOCENTER.

The only ones that do are those having pairs of opposite edges perpendicular. However, Monge, a French Mathematician, discovered a point that is similar to an orthocenter. He showed that the six planes that pass through the midpoint of an edge and are perpendicular to the opposite edge intersect in a point that is (appropriately) called the Monge point, which is shown in **Figure 6.**



FIGURE 6. TETRAHEDRON *ABCD* WITH ITS MONGE POINT.

It turns out that the Monge point has a neat property that can best be shown by example. In what follows, we will see that the plane through the midpoint of edge *AB* perpendicular to edge *DC*, the plane through the midpoint of edge *BC* perpendicular to edge *AD*, and the plane through the midpoint of edge *CD* perpendicular to edge *AB* all meet in a point.

The example in **Table 4** suggests that the centroid is the midpoint of the segment joining the Monge point and the circumcenter, as shown in **Figure 7.** This turns out to be true in general and is the three-dimensional equivalent of the Euler line for triangles in the plane. ❑



FIGURE 7. TETRAHEDRON *ABCD* WITH ITS SPECIAL SEGMENT JOINING THE MONGE POINT (*M*) TO THE CIRCUMCENTER (*R*).

Finding the Monge Point

```
AB = OB - OA; AD = OD - OA; CD = OD - OC

{-4, 4, 10}

midAB = (1/2) (OA + OB); midBC = (1/2) (OB + OC);
midCD = (1/2) (OC + OD)

{6, 0, 5}

bisectAB = CD.(OX - midAB)

-4 (-4 + x) + 4 (-6 + y) + 10 z

bisectBC = AD.(OX - midBC)

2 (-7 + x) - 2 (-3 + y) + 10 z

bisectCD = AB.(OX - midCD)

4 (-6 + x) + 4 y

Solve[{bisectAB == 0, bisectBC == 0, bisectCD == 0}, {x, y, z}]

{{x -> 3, y -> 3, z -> 4/5}}

mongept = {3, 3, 4/5}

{3, 3, 4/5}

(1/2) (mongept + OR)

{5, 3, 5/2}

centroid

{5, 3, 5/2}
```

TABLE 4.